LIBNAME Sullivan "_____"; /* Specify the directory where you want to save the output data */
/**********************************************************************************************/
/*********** SULLIVAN HEALTH EXPECTANCY USING AN UNABRIDGED LIFE TABLE : EXAMPLE 1 ************/
/**********************************************************************************************/
/******** Calculation of the life table and life expectancy ********/
PROC IMPORT OUT= WORK.input_example1 

DATAFILE= "_____\input_examples.xls" 

/* Specify the directory of the input data */
DBMS=EXCEL REPLACE;

SHEET="input_example1$"; GETNAMES=YES; MIXED=NO; SCANTEXT=YES; USEDATE=YES; SCANTIME=YES;

RUN;

DATA example1;

RETAIN x Px Dx mx qx rx pix;

SET input_example1;

mx=Dx/Px;

qx=mx/(1+(0.5*mx));

IF x=0 THEN qx=1-(1-((Dx-Dx2)/Px1))*(1-(Dx2/B));

IF x=85 THEN qx=.;

rx=1-qx;

DROP Px1 Dx2 B;

FORMAT Dx 6.0 Px 8.1 pix 5.3 mx qx rx 8.6;

RUN;

PROC EXPAND DATA=example1 OUT=example1 (DROP=time) METHOD=NONE;

CONVERT rx=rxrx / TRANSFORM=(PRODUCT);

RUN;

DATA example1;

RETAIN x Px Dx mx qx lx pix;

SET example1;

lagrxrx=LAG(rxrx);

IF x=0 THEN lx=100000;

ELSE lx=100000*lagrxrx;

DROP rx rxrx lagrxrx;

FORMAT lx 9.2;

RUN;

PROC SORT DATA=example1 OUT=example1;

BY DESCENDING x;

RUN;

DATA example1;

RETAIN x Px Dx mx qx lx LLx pix;

SET example1;

laglx=LAG(lx);

LLx=(lx+laglx)/2;

IF x=0 THEN LLx=0.2*lx+0.8*laglx;

IF x=85 THEN LLx=lx/mx;

DROP laglx;

FORMAT Llx 10.2;

RUN;

PROC EXPAND DATA=example1 OUT=example1 (DROP=time) METHOD=NONE;

CONVERT LLx=Tx / TRANSFORM=(SUM);

RUN;

PROC SORT DATA=example1 OUT=example1;

BY x;

RUN;

DATA example1;

RETAIN x Px Dx mx qx lx LLx Tx ex pix;

SET example1;

ex=Tx/lx;

FORMAT ex 4.1;

RUN;

/******** Calculation of Disability-Free Life Expectancy (DFLE) ********/
DATA example1;

SET example1;

LLx_disafr=LLx*(1-pix);

FORMAT LLx_disafr 10.2;

RUN;

PROC SORT DATA=example1 OUT=example1;

BY DESCENDING x;

RUN;

PROC EXPAND DATA=example1 OUT=example1 (DROP=time) METHOD=NONE;

CONVERT LLx_disafr=Tx_disafr / TRANSFORM=(SUM);

RUN;

PROC SORT DATA=example1 OUT=example1;

BY x;

RUN;

DATA example1;

RETAIN x Px Dx mx qx lx LLx Tx ex pix LLx_disafr Tx_disafr DFLEx percDFLEx;

SET example1;

DFLEx=Tx_disafr/lx;

percDFLEx=100*(DFLEx/ex);

FORMAT DFLEx percDFLEx 4.1 ;

RUN;

DATA Sullivan.example1;

SET example1;

LABEL x='Age'

  Px='Mid-year population'

  Dx='No. deaths'

  mx='Central Death rate'

  qx='Conditional probability of death'

  lx='Numbers surviving to age x'

  LLx='Person years lived at age x'

  Tx='Total number of years lived from age x'

  ex='Total life Expectancy'

  pix='Proportion with disability'

  LLx_disafr='Person years lived without disability at age x'

  Tx_disafr='Total years lived without disability from age x'

  DFLEx='Disability-free life expectancy'

  percDFLEx='Proportion of life spent disability-free';

RUN;

/**********************************************************************************************/
/*********** SULLIVAN HEALTH EXPECTANCY USING AN UNABRIDGED LIFE TABLE : EXAMPLE 2 ************/
/**********************************************************************************************/
/******** Calculation of the life table and life expectancy ********/
PROC FORMAT library=work;

VALUE agegr 0='0' 
   1='1-4'    5='5-9'    10='10-14' 15='15-19' 20='20-24' 25='25-29' 




30='30-34' 35='35-39' 40='40-44' 45='45-49' 50='50-54' 55='55-59' 60='60-64'



65='65-69' 70='70-74' 75='75-79' 80='80-84' 85='85+';

RUN;

DATA input_example2;

RETAIN x agegr lx LLx pix;

SET example1;

FORMAT agegr agegr.;

IF  x=0 



 THEN agegr=0;  IF (x ge 1 and x le 4) 
 THEN agegr=1;

IF (x ge 5 and x le 9) 
 THEN agegr=5;  IF (x ge 10 and x le 14) THEN agegr=10;

IF (x ge 15 and x le 19) THEN agegr=15; IF (x ge 20 and x le 24) THEN agegr=20;

IF (x ge 25 and x le 29) THEN agegr=25; IF (x ge 30 and x le 34) THEN agegr=30;

IF (x ge 35 and x le 39) THEN agegr=35; IF (x ge 40 and x le 44) THEN agegr=40;

IF (x ge 45 and x le 49) THEN agegr=45; IF (x ge 50 and x le 54) THEN agegr=50;

IF (x ge 55 and x le 59) THEN agegr=55; IF (x ge 60 and x le 64) THEN agegr=60;

IF (x ge 65 and x le 69) THEN agegr=65; IF (x ge 70 and x le 74) THEN agegr=70;

IF (x ge 75 and x le 79) THEN agegr=75; IF (x ge 80 and x le 84) THEN agegr=80;

IF (x ge 85) 


 THEN agegr=85;

KEEP x agegr lx LLx pix;

RUN;

PROC MEANS DATA=input_example2 IDMIN SUM;

VAR LLx;

BY agegr;

ID x lx pix;

OUTPUT OUT=example2 (DROP=_type_ _freq_) SUM(LLx)=nLx;

RUN;

PROC SORT DATA=example2 OUT=example2;

BY DESCENDING agegr;

RUN;

PROC EXPAND DATA=example2 OUT=example2 (DROP=time) METHOD=NONE;

CONVERT nLx=Tx / TRANSFORM=(SUM);

RUN;

PROC SORT DATA=example2 OUT=example2;

BY agegr;

RUN;

DATA example2;

RETAIN x agegr lx nLx Tx ex pix;

SET example2;

ex=Tx/lx;

FORMAT ex 4.1;

RUN;

/******** Calculation of Disability-Free Life Expectancy (DFLE) ********/
DATA example2;

SET example2;

nLx_disafr=nLx*(1-pix);

FORMAT nLx_disafr 10.2;

RUN;

PROC SORT DATA=example2 OUT=example2;

BY DESCENDING agegr;

RUN;

PROC EXPAND DATA=example2 OUT=example2 (DROP=time) METHOD=NONE;

CONVERT nLx_disafr=Tx_disafr / TRANSFORM=(SUM);

RUN;

PROC SORT DATA=example2 OUT=example2;

BY agegr;

RUN;

DATA example2;

RETAIN x agegr lx nLx Tx ex pix nLx_disafr Tx_disafr DFLEx percDFLEx;

SET example2;

DFLEx=Tx_disafr/lx;

percDFLEx=100*(DFLEx/ex);

FORMAT DFLEx percDFLEx 4.1 ;

RUN;

DATA Sullivan.example2;

RETAIN x agegroup lx nLx Tx ex pix nLx_disafr Tx_disafr DFLEx percDFLEx;

SET example2;

drop agegr;

length agegroup $ 10;

IF x=0  THEN agegroup='0';     IF x=1  THEN agegroup='1-4';   IF x=5  THEN agegroup='5-9';

IF x=5  THEN agegroup='5-9';   IF x=10 THEN agegroup='10-14'; IF x=15 THEN agegroup='15-19';

IF x=20 THEN agegroup='20-24'; IF x=25 THEN agegroup='25-29'; IF x=30 THEN agegroup='30-34';

IF x=35 THEN agegroup='35-39'; IF x=40 THEN agegroup='40-44'; IF x=45 THEN agegroup='45-49';

IF x=50 THEN agegroup='50-54'; IF x=55 THEN agegroup='55-59'; IF x=60 THEN agegroup='60-64';

IF x=65 THEN agegroup='65-69'; IF x=70 THEN agegroup='70-74'; IF x=75 THEN agegroup='75-79';

IF x=80 THEN agegroup='80-84'; IF x=85 THEN agegroup='85+';

LABEL x='Age at start of interval'

  agegroup='Age group'

  lx='Numbers surviving to age x'

  nLx='Person years lived in age interval'

  Tx='Total number of years lived from age x'

  ex='Total life Expectancy'

  pix='Proportion with disability'

  nLx_disafr='Person years lived without disability in age interval'

  Tx_disafr='Total years lived without disability from age x'

  DFLEx='Disability-free life expectancy'

  percDFLEx='Proportion of life spent disability-free';

RUN;

/**********************************************************************************************/
/************ SULLIVAN HEALTH EXPECTANCY USING AN ABRIDGED LIFE TABLE : EXAMPLE 3 *************/
/**********************************************************************************************/
/******** Calculation of the life table and life expectancy ********/
DATA input_example3;

RETAIN x agegr Px Dx Px1 Dx2 B pix; 

SET input_example1;

FORMAT agegr agegr. Px 8.1 Dx 6.0 pix 5.3 ;

IF  x=0 



 THEN agegr=0;  IF (x ge 1 and x le 4) 
 THEN agegr=1;

IF (x ge 5 and x le 9) 
 THEN agegr=5;  IF (x ge 10 and x le 14) THEN agegr=10;

IF (x ge 15 and x le 19) THEN agegr=15; IF (x ge 20 and x le 24) THEN agegr=20;

IF (x ge 25 and x le 29) THEN agegr=25; IF (x ge 30 and x le 34) THEN agegr=30;

IF (x ge 35 and x le 39) THEN agegr=35; IF (x ge 40 and x le 44) THEN agegr=40;

IF (x ge 45 and x le 49) THEN agegr=45; IF (x ge 50 and x le 54) THEN agegr=50;

IF (x ge 55 and x le 59) THEN agegr=55; IF (x ge 60 and x le 64) THEN agegr=60;

IF (x ge 65 and x le 69) THEN agegr=65; IF (x ge 70 and x le 74) THEN agegr=70;

IF (x ge 75 and x le 79) THEN agegr=75; IF (x ge 80 and x le 84) THEN agegr=80;

IF (x ge 85) 


 THEN agegr=85;

RUN;

PROC MEANS DATA=input_example3 IDMIN SUM;

VAR Px Dx;

BY agegr;

ID x Px1 Dx2 B pix;

OUTPUT OUT=example3 (DROP=_type_ _freq_) SUM(Px)=Px SUM(Dx)=Dx;

RUN;

DATA example3;

RETAIN x agegr Px Dx nmx ax nqx nrx pix;

SET example3;

nmx=Dx/Px;

ax=0.5;

nqx=(5*nmx)/(1+((1-ax)*5*nmx));

IF agegr=0 THEN nqx=1-(1-((Dx-Dx2)/Px1))*(1-(Dx2/B));

IF agegr=1 THEN nqx=(4*nmx)/(1+((1-ax)*4*nmx));

IF agegr=85 THEN nqx=.;

nrx=1-nqx;

DROP Dx2 Px1 B;

FORMAT nmx 8.6 nqx 10.8;

RUN;

PROC EXPAND DATA=example3 OUT=example3 (DROP=time) METHOD=NONE;

CONVERT nrx=nrxrx / TRANSFORM=(PRODUCT);

RUN;

DATA example3;

RETAIN x agegr Px Dx nmx ax nqx lx pix;

SET example3;

lagnrxrx=LAG(nrxrx);

IF agegr=0 THEN lx=100000;

ELSE lx=100000*lagnrxrx;

DROP nrx nrxrx lagnrxrx;

FORMAT lx 8.1;

RUN;

PROC SORT DATA=example3 OUT=example3;

BY DESCENDING agegr;

RUN;

DATA example3;

RETAIN x agegr Px Dx nmx ax nqx lx nLx pix;

SET example3;

laglx=LAG(lx);

nLx=((lx*5*ax)+(laglx*5*(1-ax)));

IF agegr=0 THEN nLx=((lx*0.2)+(laglx*0.8));

IF agegr=1 THEN nLx=((lx*4*ax)+(laglx*4*(1-ax)));

IF agegr=85 THEN nLx=lx/nmx;

DROP laglx;

FORMAT nLx 9.1;

RUN;

PROC EXPAND DATA=example3 OUT=example3 (DROP=time) METHOD=NONE;

CONVERT nLx=Tx / TRANSFORM=(SUM);

RUN;

PROC SORT DATA=example3 OUT=example3;

BY agegr;

RUN;

DATA example3;

RETAIN x agegr Px Dx nmx ax nqx lx nLx Tx ex pix;

SET example3;

ex=Tx/lx;

FORMAT ex 4.1;

RUN;

/******** Calculation of Disability-Free Life Expectancy (DFLE) ********/
DATA example3;

SET example3;

nLx_disafr=nLx*(1-pix);

FORMAT nLx_disafr 9.1;

RUN;

PROC SORT DATA=example3 OUT=example3;

BY DESCENDING agegr;

RUN;

PROC EXPAND DATA=example3 OUT=example3 (DROP=time) METHOD=NONE;

CONVERT nLx_disafr=Tx_disafr / TRANSFORM=(SUM);

RUN;

PROC SORT DATA=example3 OUT=example3;

BY agegr;

RUN;

DATA example3;

RETAIN x agegr Px Dx nmx ax nqx lx nLx Tx ex pix nLx_disafr Tx_disafr DFLEx percDFLEx;

SET example3;

DFLEx=Tx_disafr/lx;

percDFLEx=100*(DFLEx/ex);

FORMAT DFLEx percDFLEx 4.1 ;

RUN;

DATA Sullivan.example3;

RETAIN x agegroup Px Dx nmx ax nqx lx nLx Tx ex pix nLx_disafr Tx_disafr DFLEx percDFLEx;

SET example3;

drop agegr;

length agegroup $ 10;

IF x=0  THEN agegroup='0';     IF x=1  THEN agegroup='1-4';   IF x=5  THEN agegroup='5-9';

IF x=5  THEN agegroup='5-9';   IF x=10 THEN agegroup='10-14'; IF x=15 THEN agegroup='15-19';

IF x=20 THEN agegroup='20-24'; IF x=25 THEN agegroup='25-29'; IF x=30 THEN agegroup='30-34';

IF x=35 THEN agegroup='35-39'; IF x=40 THEN agegroup='40-44'; IF x=45 THEN agegroup='45-49';

IF x=50 THEN agegroup='50-54'; IF x=55 THEN agegroup='55-59'; IF x=60 THEN agegroup='60-64';

IF x=65 THEN agegroup='65-69'; IF x=70 THEN agegroup='70-74'; IF x=75 THEN agegroup='75-79';

IF x=80 THEN agegroup='80-84'; IF x=85 THEN agegroup='85+';

LABEL x='Age at start of interval'

  agegroup='Age group'

  Px='Mid-year population'

  Dx='No. deaths'

  nmx='Central Death rate'

  nqx='Conditional probability of death in age interval'

  lx='Numbers surviving to age x'

  nLx='Person years lived in age interval'

  Tx='Total number of years lived from age x'

  ex='Total life Expectancy'

  pix='Proportion with disability'

  nLx_disafr='Person years lived without disability in age interval'

  Tx_disafr='Total years lived without disability from age x'

  DFLEx='Disability-free life expectancy'

  percDFLEx='Proportion of life spent disability-free';  

RUN;

/**********************************************************************************************/
/************* THE STANDARD ERROR OF THE SULLIVAN HEALTH EXPECTANCY : EXAMPLE 4 ***************/
/**********************************************************************************************/
/******** Approximate standard errors ignoring the variance of the mortality rates ********/
PROC IMPORT OUT= WORK.input_example4 

DATAFILE= "_____\input_examples.xls" 
   /* Specify the directory of the input data */
DBMS=EXCEL REPLACE;

SHEET="input_example4$"; GETNAMES=YES; MIXED=NO; SCANTEXT=YES; USEDATE=YES; SCANTIME=YES;

RUN;

DATA example4A;

RETAIN x agegr ax Dx nqx lx nLx ex pix nLx_disafr Tx_disafr DFLEx Nx S2_pix L2S2_pix;

merge example3 input_example4;

BY agegr;

S2_pix=(pix*(1-pix))/Nx;

L2S2_pix=(nLx**2)*S2_pix;

KEEP x agegr ax Dx nqx lx nLx ex pix nLx_disafr Tx_disafr DFLEx Nx S2_pix L2S2_pix;

FORMAT S2_pix 8.6 L2S2_pix 9.0;

RUN;

PROC SORT DATA=example4A OUT=example4A;

BY DESCENDING agegr;

RUN;

PROC EXPAND DATA=example4A OUT=example4A (DROP=time) METHOD=NONE;

CONVERT L2S2_pix=T_L2S2_pix/TRANSFORM=(SUM);

RUN;

PROC SORT DATA=example4A OUT=example4A;

BY agegr;

RUN;

DATA example4A;

RETAIN x agegr ax Dx nqx lx nLx ex pix nLx_disafr Tx_disafr DFLEx Nx S2_pix L2S2_pix 


   T_L2S2_pix S2_DFLEx S_DFLEx;

SET example4A;

S2_DFLEx=T_L2S2_pix/(lx**2);

S_DFLEx=sqrt(S2_DFLEx);

FORMAT S2_pix 8.6 L2S2_pix T_L2S2_pix 10.0 S2_DFLEx 7.5 S_DFLEx 5.3;

RUN;

DATA Sullivan.example4A;

RETAIN x agegroup lx nLx ex pix nLx_disafr Tx_disafr DFLEx Nx S2_pix L2S2_pix T_L2S2_pix 


   S2_DFLEx S_DFLEx;

SET example4A;

drop agegr;

length agegroup $ 10;

IF x=0  THEN agegroup='0';     IF x=1  THEN agegroup='1-4';   IF x=5  THEN agegroup='5-9';

IF x=5  THEN agegroup='5-9';   IF x=10 THEN agegroup='10-14'; IF x=15 THEN agegroup='15-19';

IF x=20 THEN agegroup='20-24'; IF x=25 THEN agegroup='25-29'; IF x=30 THEN agegroup='30-34';

IF x=35 THEN agegroup='35-39'; IF x=40 THEN agegroup='40-44'; IF x=45 THEN agegroup='45-49';

IF x=50 THEN agegroup='50-54'; IF x=55 THEN agegroup='55-59'; IF x=60 THEN agegroup='60-64';

IF x=65 THEN agegroup='65-69'; IF x=70 THEN agegroup='70-74'; IF x=75 THEN agegroup='75-79';

IF x=80 THEN agegroup='80-84'; IF x=85 THEN agegroup='85+';

LABEL x='Age at start of interval'

  agegroup='Age group'

  Dx='No. deaths'

  nqx='Conditional probability of death in age interval'

  lx='Numbers surviving to age x'

  nLx='Person years lived in age interval'

  ex='Total life Expectancy'

  pix='Proportion with disability'

  nLx_disafr='Person years lived without disability in age interval'

  Tx_disafr='Total years lived without disability from age x'

  DFLEx='Disability-free life expectancy'

  Nx='Number in survey in age interval'

  S2_pix='Variance of prevalence disability-free'

  S2_DFLEx='Variance of DFLEx'
      S_DFLEx='Standard error of DFLEx';  

KEEP x agegroup lx nLx ex pix nLx_disafr Tx_disafr DFLEx Nx S2_pix L2S2_pix T_L2S2_pix 


 S2_DFLEx S_DFLEx;

RUN;

/******** Standard errors taking into account the variance of the mortality rates ********/
DATA example4B;

SET example4A;

S2_DFLEx_1=S2_DFLEx;

S2_qx=((nqx**2)*(1-nqx))/Dx;

IF agegr=85 THEN S2_qx=0;

KEEP x agegr ax Dx nqx lx pix DFLEx S2_DFLEx_1 S2_qx;

FORMAT S2_DFLEx_1 7.5 S2_qx 10.8;

RUN;

PROC SORT DATA=example4B OUT=example4B;

BY DESCENDING agegr;

RUN;

DATA example4B;

SET example4B;

lagDFLEx=LAG(DFLEx);

RUN;

PROC SORT DATA=example4B OUT=example4B;

BY agegr;

RUN;

DATA example4B;

SET example4B;

y=(((1-ax)*5*(1-pix))+lagDFLEx);

IF agegr=0 THEN y=(((1-ax)*(1-pix))+lagDFLEx);

IF agegr=1 THEN y=(((1-ax)*4*(1-pix))+lagDFLEx);

IF agegr=85 THEN y=0;

z=((y**2)*(lx**2)*(S2_qx));

DROP lagDFLEx;

FORMAT y 7.4 z 12.3; 

RUN;

PROC SORT DATA=example4B;

BY DESCENDING agegr;

RUN;

PROC EXPAND DATA=example4B OUT=example4B (DROP=time) METHOD=NONE;

CONVERT z=T_z / TRANSFORM=(SUM);

RUN;

PROC SORT DATA=example4B OUT=example4B;

BY agegr;

RUN;

DATA example4B;

RETAIN x agegr ax Dx nqx lx pix DFLEx S2_DFLEx_1 S2_qx y z T_z S2_DFLEx_2 S2_DFLEx;

SET example4B;

S2_DFLEx_2=T_z/(lx**2);

S2_DFLEx=S2_DFLEx_1+S2_DFLEx_2;

FORMAT S2_DFLEx_2 S2_DFLEx 7.5;

RUN;

DATA Sullivan.example4B;

RETAIN x agegroup ax Dx nqx lx pix DFLEx S2_DFLEx_1 S2_qx y z T_z S2_DFLEx_2 S2_DFLEx;

SET example4B;

drop agegr;

length agegroup $ 10;

IF x=0  THEN agegroup='0';     IF x=1  THEN agegroup='1-4';   IF x=5  THEN agegroup='5-9';

IF x=5  THEN agegroup='5-9';   IF x=10 THEN agegroup='10-14'; IF x=15 THEN agegroup='15-19';

IF x=20 THEN agegroup='20-24'; IF x=25 THEN agegroup='25-29'; IF x=30 THEN agegroup='30-34';

IF x=35 THEN agegroup='35-39'; IF x=40 THEN agegroup='40-44'; IF x=45 THEN agegroup='45-49';

IF x=50 THEN agegroup='50-54'; IF x=55 THEN agegroup='55-59'; IF x=60 THEN agegroup='60-64';

IF x=65 THEN agegroup='65-69'; IF x=70 THEN agegroup='70-74'; IF x=75 THEN agegroup='75-79';

IF x=80 THEN agegroup='80-84'; IF x=85 THEN agegroup='85+';

LABEL x='Age at start of interval'

  agegroup='Age group'

  Dx='No. deaths'

  nqx='Conditional probability of death in age interval'

  lx='Numbers surviving to age x'

  pix='Proportion with disability'

  DFLEx='Disability-free life expectancy'

  S2_qx='Variance of probability of death'

  S2_DFLEx_1='Variance of DFLEx (due to prevalence)'

  S2_DFLEx_2='Variance of DFLEx (due to mortality)'

  S2_DFLEx='Total variance of DFLEx';

RUN;

/**********************************************************************************************/
/*********** TESTING THE EQUALITY OF TWO SULLIVAN HEALTH EXPECTANCIES : EXAMPLE 5 *************/
/**********************************************************************************************/
PROC IMPORT OUT= WORK.input_example5 

DATAFILE= "_____\input_examples.xls" 

/* Specify the directory of the input data */
DBMS=EXCEL REPLACE;

SHEET="input_example5$"; GETNAMES=YES; MIXED=NO; SCANTEXT=YES; USEDATE=YES; SCANTIME=YES;

RUN;

DATA example5;

RETAIN x agegr DFLEf S_DFLEf DFLEm S_DFLEm diff_DFLE S_diff_DFLE z p;

SET input_example5;

x=agegr;

diff_DFLE=DFLEf-DFLEm;

S_diff_DFLE=S_DFLEf+S_DFLEm;

z=diff_DFLE/S_diff_DFLE;

pvalue=2*(1-cdf('NORMAL',z));

length p $ 7;

IF pvalue>0.20 THEN p='>0.20';

IF pvalue<0.20 THEN p='<0.20';

IF pvalue<0.10 THEN p='<0.10';

IF pvalue<0.05 THEN p='<0.05';

IF pvalue<0.02 THEN p='<0.02';

IF pvalue<0.01 THEN p='<0.01';

IF pvalue<0.002 THEN p='<0.002';

IF pvalue<0.001 THEN p='<0.001';

DROP pvalue;

FORMAT agegr agegr. DFLEf DFLEm 4.1 S_DFLEf S_DFLEm diff_DFLE S_diff_DFLE z 4.2;

RUN;

DATA Sullivan.example5;

RETAIN x agegroup DFLEf S_DFLEf DFLEm S_DFLEm diff_DFLE S_diff_DFLE z p;

SET example5;

drop agegr;

length agegroup $ 10;

IF x=0  THEN agegroup='0';     IF x=1  THEN agegroup='1-4';   IF x=5  THEN agegroup='5-9';

IF x=5  THEN agegroup='5-9';   IF x=10 THEN agegroup='10-14'; IF x=15 THEN agegroup='15-19';

IF x=20 THEN agegroup='20-24'; IF x=25 THEN agegroup='25-29'; IF x=30 THEN agegroup='30-34';

IF x=35 THEN agegroup='35-39'; IF x=40 THEN agegroup='40-44'; IF x=45 THEN agegroup='45-49';

IF x=50 THEN agegroup='50-54'; IF x=55 THEN agegroup='55-59'; IF x=60 THEN agegroup='60-64';

IF x=65 THEN agegroup='65-69'; IF x=70 THEN agegroup='70-74'; IF x=75 THEN agegroup='75-79';

IF x=80 THEN agegroup='80-84'; IF x=85 THEN agegroup='85+';

LABEL x='Age at start of interval'

  agegroup='Age group'

  DFLEf='Disability-free life expectancy females'
      S_DFLEf='Standard error of DFLE females'
      DFLEm='Disability-free life expectancy males'
      S_DFLEm='Standard error of DFLE males'
      diff_DFLE='Difference in DFLE between males and females'
      S_diff_DFLE='Approximate standard error of difference in DFLE between males and females'

  z='z statistic'

  p='p value';

RUN;

/**********************************************************************************************/
/** SULLIVAN HEALTH EXPECTANCY USING SEPARATE DATA SOURCES FOR INSTITUTIONALISED : EXAMPLE 6 **/
/**********************************************************************************************/
/******** Calculation of Disability-Free Life Expectancy (DFLE) ********/
PROC IMPORT OUT= WORK.input_example6 

DATAFILE= "_____\input_examples.xls" 

/* Specify the directory of the input data */
DBMS=EXCEL REPLACE;

SHEET="input_example6$"; GETNAMES=YES; MIXED=NO; SCANTEXT=YES; USEDATE=YES; SCANTIME=YES;

RUN;

DATA example6A;

RETAIN x agegr lx nLx ex pix Ix pix_total nLx_disafr_total DFLEx;

merge input_example6 example3;

pix_total=((1-Ix)*(pix))+Ix;

nLx_disafr_total=nLx*(1-pix_total);

KEEP x agegr lx nLx ex pix Ix pix_total nLx_disafr_total DFLEx;

FORMAT Ix pix_total 5.3 nLx_disafr_total 9.1;

RUN;

PROC SORT DATA=example6A OUT=example6A;

BY DESCENDING agegr;

RUN;

PROC EXPAND DATA=example6A OUT=example6A (DROP=time) METHOD=NONE;

CONVERT nLx_disafr_total=Tx_disafr_total/TRANSFORM=(SUM);

RUN;

PROC SORT DATA=example6A OUT=example6A;

BY agegr;

RUN;

DATA example6A;

RETAIN x agegr lx nLx ex pix Ix pix_total nLx_disafr_total Tx_disafr_total DFLEx_total DFLEx;

SET example6A;

DFLEx_total=Tx_disafr_total/lx;

FORMAT DFLEx_total 4.1;

RUN;

DATA Sullivan.example6A;

RETAIN x agegroup lx nLx ex pix Ix pix_total nLx_disafr_total Tx_disafr_total DFLEx_total DFLEx;

SET example6A;

drop agegr;

length agegroup $ 10;

IF x=0  THEN agegroup='0';     IF x=1  THEN agegroup='1-4';   IF x=5  THEN agegroup='5-9';

IF x=5  THEN agegroup='5-9';   IF x=10 THEN agegroup='10-14'; IF x=15 THEN agegroup='15-19';

IF x=20 THEN agegroup='20-24'; IF x=25 THEN agegroup='25-29'; IF x=30 THEN agegroup='30-34';

IF x=35 THEN agegroup='35-39'; IF x=40 THEN agegroup='40-44'; IF x=45 THEN agegroup='45-49';

IF x=50 THEN agegroup='50-54'; IF x=55 THEN agegroup='55-59'; IF x=60 THEN agegroup='60-64';

IF x=65 THEN agegroup='65-69'; IF x=70 THEN agegroup='70-74'; IF x=75 THEN agegroup='75-79';

IF x=80 THEN agegroup='80-84'; IF x=85 THEN agegroup='85+';

LABEL x='Age at start of interval'

  agegroup='Age group'

  lx='Numbers surviving to age x'

  nLx='Person years lived in age interval'

  ex='Total life Expectancy'
 
  pix='Proportion of age group in survey with disability'

  Ix='Proportion in institutions'

  pix_total='Proportion of age group (total) with disability'

  nLx_disafr_total='Person years lived without disability in age interval'

  Tx_disafr_total='Total years lived without disability from age x'

  DFLEx_total='Disability-free life expectancy (including inst)'

  DFLEx='Disability-free life expectancy (excluding inst)';

RUN;

/******** Approximate standard errors ignoring the variance of the mortality rates ********/
DATA example6B;

merge example6A input_example4;

BY agegr;

S2_pix_total=(1-Ix)*(1-Ix)*pix_total*(1-pix_total)/Nx;

L2S2_pix_total=(nLx**2)*S2_pix_total;

KEEP x agegr lx nLx ex pix Ix pix_total DFLEx_total Nx S2_pix_total L2S2_pix_total;

FORMAT S2_pix_total 8.6 L2S2_pix_total 10.0;

RUN;

PROC SORT DATA=example6B OUT=example6B;

BY DESCENDING agegr;

RUN;

PROC EXPAND DATA=example6B OUT=example6B (DROP=time) METHOD=NONE;

CONVERT L2S2_pix_total=T_L2S2_pix_total/TRANSFORM=(SUM);

RUN;

PROC SORT DATA=example6B OUT=example6B;

BY agegr;

RUN;

DATA example6B;

RETAIN x agegr lx nLx ex pix Ix pix_total DFLEx_total Nx S2_pix_total L2S2_pix_total 


   T_L2S2_pix_total S2_DFLEx_total S_DFLEx_total;

set example6B;

S2_DFLEx_total=T_L2S2_pix_total/(lx**2);

S_DFLEx_total=sqrt(S2_DFLEx_total);

FORMAT S2_DFLEx_total 7.5 S_DFLEx_total 5.3;

RUN;

DATA Sullivan.example6B;

RETAIN x agegroup lx nLx ex pix Ix pix_total DFLEx_total Nx S2_pix_total L2S2_pix_total 


   T_L2S2_pix_total S2_DFLEx_total S_DFLEx_total;

SET example6B;

drop agegr;

length agegroup $ 10;

IF x=0  THEN agegroup='0';     IF x=1  THEN agegroup='1-4';   IF x=5  THEN agegroup='5-9';

IF x=5  THEN agegroup='5-9';   IF x=10 THEN agegroup='10-14'; IF x=15 THEN agegroup='15-19';

IF x=20 THEN agegroup='20-24'; IF x=25 THEN agegroup='25-29'; IF x=30 THEN agegroup='30-34';

IF x=35 THEN agegroup='35-39'; IF x=40 THEN agegroup='40-44'; IF x=45 THEN agegroup='45-49';

IF x=50 THEN agegroup='50-54'; IF x=55 THEN agegroup='55-59'; IF x=60 THEN agegroup='60-64';

IF x=65 THEN agegroup='65-69'; IF x=70 THEN agegroup='70-74'; IF x=75 THEN agegroup='75-79';

IF x=80 THEN agegroup='80-84'; IF x=85 THEN agegroup='85+';

LABEL x='Age at start of interval'

  agegroup='Age group'

  lx='Numbers surviving to age x'

  nLx='Person years lived in age interval'

  ex='Total life Expectancy'
 
  pix='Proportion of age group in survey with disability'

  Ix='Proportion in institutions'

  pix_total='Proportion of age group (total) with disability'

  DFLEx_total='Disability-free life expectancy (including inst)'

  Nx='Number in survey in age interval'

  S2_pix_total='Variance of prevalence disability-free'

  S2_DFLEx_total='Variance of DFLEx'
      S_DFLEx_total='Standard error of DFLEx';  

RUN;
